



Bridge of Life
Education

SOC Design Verilog Delay Explained

Jiin Lai


Topics

1. Propagation Delay
2. Inertial / Transport Delay
3. Blocking / nonblocking
4. LHS (Inter) / RHS (Intra) delay
5. Procedure block / Continuous assignment


Use the right delay model

Switching Waveform


$(R_{dh} * C_{load})$ is called the **RC time constant**



$$V = V_{dd} * [1 - e^{-t / (R_{dh} * C_{load})}]$$




$$V = V_{dd} * e^{-t / (R_{dl} * C_{load})}$$






Switching from high to low

- When switching from high to low, **PMOS pull-up and the NMOS pull-down transistors are both on simultaneously for a short moment.**
- Later, the pull-up structure turns off the current flow from C_{load} to V_{ss} .
- **After the output reaches the final state, there is no current flow** as the capacitance C_{load} is completely discharged.

Cell is switching
(pullup, pull-down both on)






Discharging to logic-0
(pull-up off, pull-down on)



Propagation Delay

- Timing measurement on the switching waveforms.
 - Threshold point of an input falling edge: ``input_threshold_pct_fall : 50.0;`
 - Threshold point of an input rising edge: ``input_threshold_pct_rise : 50.0;`
 - Threshold point of an output falling edge: ``output_threshold_pct_fall : 50.0;`
 - Threshold point of an output rising edge: ``output_threshold_pct_rise : 50.0;`
- Propagation Delay:
 - Output fall delay (***T_f***)
 - Output rise delay (***T_r***)



Inertial / Transport delay

- **Inertial delay** models only propagate signals to an output with pulse width greater than its propagation delay. If the pulse width is less, the early edge scheduled is replaced by the later edge. (Default)
 - **+pulse_r/100 +pulse_e/100**
- **Transport delay** models propagate all signals to an output after any input signals change. Scheduled output value changes are queued for transport delay models. (Used for behavior model)
 - **+pulse_r/0 +pulse_e/0**
- **Semi-realistic delay**
 - **+pulse_r/30 +pulse_e/70**
reject pulses less than 30%,
propagate unknowns for pulses between 30-70% and
pass all pulses greater than 70% of propagation delay.

Inter (LHS) / Intra (RHS) Delay

Inter (LHS) Delay **#<delay> <LHS> = <RHS>**

The statement is executed after the delay. Most commonly used.

Intra (RHS) Delay **<LHS> = #<delay> <RHS>**

The statement is evaluated with the signals on the RHS captured first. And the assignment to the LHS signal after the delay.

Category

- Blocking / nonblocking
- LHS (Inter) / RHS (Intra) delay
- Procedure block / Continuous assignment
- Inertial / transport delay

	Blocking/RHS	Blocking/LHS	Nonblocking/RHS	Nonblocking/LHS
Procedure Block	X	For Testbench	Transport Delay	Not for Comb Logic
Continuous Assignment	Transport Delay ?	Inertial Delay	X	X

Procedure Block – Blocking - LHS (Inter) Delay

Delay both the evaluation and update


Modeling Guideline: do not place delays on the LHS of blocking assignments to model combinational logic. This is a bad coding style.

Testbench Guideline: placing delays on the LHS of blocking assignments in a testbench is reasonable since the delay is just being used to time-space sequential input stimulus events.

```
module adder_t1 (co, sum, a, b, ci);  
    output      co;  
    output [3:0] sum;  
    input  [3:0] a, b;  
    input      ci;  
    reg       co;  
    reg [3:0] sum;  
  
    always @(a or b or ci)  
        #12 {co, sum} = a + b + ci;  
endmodule
```

```
always @(a or b or ci) begin  
    tmp      = a + b + ci;  
    #12 {co, sum} = tmp;  
end
```

Used for testbench



Procedure Block – Blocking - RHS (Intra) Delay

Delays the evaluation but not the update.

Delays the update but not the evaluation

Modeling Guideline: do not place delays on the RHS of blocking assignments to model combinational logic.

This is a bad coding style.

Testbench Guideline: do not place delays on the RHS of blocking assignments in a testbench.

General Guideline: placing a delay on the RHS of any blocking assignment is both confusing and a poor coding style. **This Verilog coding practice should be avoided.**

```
module adder_t6 (co, sum, a, b, ci);  
    output        co;  
    output [3:0]  sum;  
    input  [3:0]  a, b;  
    input         ci;  
    reg         co;  
    reg [3:0]    sum;
```

```
tmp      = a + b + ci;  
{co, sum} = #12 tmp;
```

```
tmp      = #12 a + b + ci;  
{co, sum} = tmp;
```

```
always @(a or b or ci)  
    {co, sum} = #12 a + b + ci;  
endmodule
```

Do not use it

Procedure Block – Nonblocking - LHS (Inter) Delay

Delay both the evaluation and the update.

Modeling Guideline: do not place delays on the LHS of nonblocking assignments to model combinational logic. This is a bad coding style

Testbench Guideline: nonblocking assignments are less efficient to simulate than blocking assignments;

```
module adder_t2 (co, sum, a, b, ci);
    output      co;
    output [3:0] sum;
    input  [3:0] a, b;
    input      ci;
    reg      co;
    reg [3:0] sum;

    always @(a or b or ci)
        #12 {co, sum} <= a + b + ci;
endmodule
```

**Do not use it for
Combinational Logic**

Procedure Block / Nonblocking / RHS (Intra)

models combinational logic with transport delays. for delay-line logic

Delay the update but not the evaluation.

Modeling Guideline: place delays on the RHS of nonblocking assignments only when trying to **model transport output-propagation** behavior. This coding style will accurately model delay lines and combinational logic with pure transport delays; however, this coding style generally causes slower simulations. **Commonly use to model clock-to-output behavior on sequential logic.**

Testbench Guideline: This coding style is used in testbench when stimulus must be scheduled on future clock edges or after a set delay, while not blocking the assignment of subsequent stimulus events in the same procedural block.


```
module adder_t3 (co, sum, a, b, ci);
    output        co;
    output [3:0]  sum;
    input  [3:0]  a, b;
    input         ci;
    reg          co;
    reg  [3:0]    sum;

    always @(a or b or ci)
        {co, sum} <= #12 a + b + ci;
endmodule
```

tmp needs to be in sensitivity list

```
always @(a or b or ci or tmp) begin
    tmp    <= #12 a + b + ci;
    {co, sum} <= tmp;
end
```

```
always @(a or b or ci or tmp) begin
    tmp    <= a + b + ci;
    {co, sum} <= #12 tmp;
end
```




Continuous assignment with delays

`assign y = ~a;` /


Continuous
assignment - no delay

`assign #5 y = ~a;` /

Continuous
assignment - LHS delay

 `assign y = #5 ~a;` /

Illegal continuous
assignment - RHS delay

 `assign y <= ~a;` /

Illegal continuous
nonblocking assignment

Continuous assignment – Blocking – LHS


Adding delays to continuous assignments accurately models combinational logic with inertial delays and is a recommended coding style.

```
module adder_t4 (co, sum, a, b, ci);  
    output      co;  
    output [3:0] sum;  
    input  [3:0] a, b;  
    input      ci;  
  
    assign #12 {co, sum} = a + b + ci;  
endmodule
```

```
assign      tmp      = a + b + ci;  
assign #12 {co, sum} = tmp;
```

```
assign #12 tmp      = a + b + ci;  
assign      {co, sum} = tmp;
```

Inertial Delay for Combinational Logic



Model Complex Combination logic with Mixed no-delay always blocks and delayed continuous assignments

Modeling Guideline: Use continuous assignments with delays to model simple combinational logic. This coding style will accurately model combinational logic with inertial delays.

Modeling Guideline: Use always blocks with no delays to model complex combinational logic using behavioral constructs such as "case-casez-casex", "if-else", etc. Then, it is driven to output with continuous assignment with inertial delay.

```
module adder_t5 (co, sum, a, b, ci);
    output      co;
    output [3:0] sum;
    input  [3:0] a, b;
    input      ci;
    reg      [4:0] tmp;

    always @(a or b or ci) begin
        tmp = a + b + ci;
    end

    assign #12 {co, sum} = tmp;
endmodule
```